
QL Retro Gaming 

The QL design was aimed at the business world and as a gaming platform I fear 

in its short production life was always to reside in the shadow of the Spectrum. 

Not to say that a large number of games were developed by enthusiasts 

experimenting with the SuperBASIC interpreter my own included.  

 

QBITS  

My contribution was a short list of programmes mostly written between 1987 

and 1990. Becoming a member of QUANTA introduced me to shows and 

gatherings and in the early nineties I was persuaded to offer some of my games 

to other members under the banner of QBITS. 

 

What follows are what I refer to as my early learning phase. That is to say being 

a father and having an interest in child education I set about writing a few 

simple programs to help them in that respect.  

 

Clocks_Tables  

Having just started infant’s school, I spent some hours one Sunday morning 

with my daughter getting her to read different times form a cardboard clock 

face. Moving the hours and minutes hands presented no problem, she did well, 

but the second hand was something of a mystery.  

 

I asked her what seconds were, several frowns and thoughtful expressions 

danced across her face. After a long pause, her face suddenly lit up. she replied, 

‘when its dinner time at school you sometimes get a second helping of pudding’. 

 

Clocks was my attempt to help understanding the analogue clock face as 

opposed to that shown on a digital clock covering both twelve and twenty four 

hour cycles. Tables was as you’d expect simple arithmetic aimed at improving 

the learning and  use of the times tables 1 to 12. 

SpellBear 

The origins of Hangman the word game are obscure, but mostly likely began in 

Victorian times. The word represented by a line of dashes for letters are guessed 

at and completed or Not as the case may be within so many turns. The Gallows 

with hanged man drawn in body parts each time a letter is incorrect. There are 

many variants a common alternative for a young child is to draw apples on a 

tree, removing them for each incorrect letter.  

 

For me I chose a bear flying high holding a set of balloons. An incorrect letter 

and the bear floated earthwards losing his balloons eventually dropping into a 

pond of water. 

 

Conundrum  

Wanting to take this to another level Conundrum was inspired by the popular 

UK TV program – Countdown, another game involving word play. The display 

was going to be somewhat obvious, a clock face with a countdown second hand, 

then a row of jumbled letters to be sorted into the correct order.  

 

How was this to be done, it seemed logical for a second row to be set below 

with the same number of underlines to represent the missing letters. The player 

would then attempt to type in letters in the correct order before the clock 

countdown and so score maximum points. For scoring, two score lines one 

showing the accrued maximum points that could have been won, the other to 

show the player’s accrued points.  

 

An added option at this point was TIME to allow the increase or decrease of the 

countdown second hand. An adjustable variable used to create more tension 

dependent on an individual’s ability. 

 

It would be heavy going having constantly to change an embedded word list 

within the Conundrum program. So what was needed next was a method of 

expanding and developing future word lists for the Conundrum game.   

 

The main requirement was going to be a line editor, one that would be capable 

of adding, inserting and deleting characters within a line of set length. At this 

point a further idea to expand to the capabilities of the Conundrum Game was to 

use word clues. Instead of showing the jumbled up letters, an option would be to 

replace them by coloured block. The Conundrum game would now have options 

to switch Clues and/or Word options On/Off.  

Conundrum Options   Clue  Load  Play  Time  Word 

 

Clue    Off/On  Default Off - Clue only shown if available.  

Load Select drive and then Wrd_filename from a list presented.  

Play Activate a word, then play against countdown 

Time  Default 30 seconds - allow countdown to be increased or decreased.  

Word     On/Off  Default On - cannot turn Off unless clue turned On 

 

QBWord_Gen  

This accompanying program to Conundrum was the outcome in generating the 

necessary word and clue lists.  A simple two column layout, typed in words on 

the left with any clue definitions to be placed on the right.  

 

QBWord_Gen Options   New  Edit  Find  Load  Save  

 

New  

Maximum number of Words and Clues 96.  

Set out as 6 pages with 16 Words / Clues to a page   

Maximum Word length 15 characters   

Maximum Clues length 32 characters including spaces 

 

Edit  

Edit the line editor. Tab between word and Clue, also up/down cursors to work 

within the page. To move along the edit line use of left/right cursors. Add to end 

of line, insert or delete a character within the line at the cursor position with any 

subsequent overspill being automatically deleted from end of line. 

 

Find  

Select a page and any line within the page.  

 

Load   Select drive device, then choose from Wrd_filenames presented 

Save Select drive device, then type name i.e. flp1_Wrd_filename: Save 

100 REMark Conundrum (QBITS •  1992) 

102 : 

104 Init:Title:Clk_Face:Menu 

 

108 DEFine PROCedure Init 

110 WTV:MODE 8 
112 OPEN #3,scr_480x230a16x4 :PAPER #3,32:CLS #3:CSIZE #3,2,0 

114 OPEN #4,con_444x32a34x220:PAPER #4,0:CLS #4:BORDER #4,1,2 

116 OPEN #5,scr_444x170a34x32:PAPER #5,1:CLS #5:BORDER #5,2,5 
118 DIM Title$(10),Command$(30),key$(16),str$(36),k$(1) 

120 DIM Sort$(16),Con1$(16),Con2$(16),DR$(6,5),SDR$(5) 

122 INK #4,7:RESTORE :FOR i=1 TO 6:READ DR$(i) 

124 DATA 'mdv1_','mdv2_','flp1_','flp2_','ram1_','ram2_' 

126 max_score%=0:per_score%=0:count%=0:D%=3:SDR$=DR$(3):key$='' 

128 END DEFine 

 

132 DEFine PROCedure Title 

134 Title$='Conundrum':BLOCK #3,480,15,0,0,0 
136 BLOCK #3,180,22,156,6,0:BLOCK #3,176,22,156,2,4 

138 CSIZE #3,3,1:OVER #3,1 

140 INK #3,0:FOR i=0 TO 3:CURSOR #3,176-i,2  :PRINT #3,Title$ 
142 INK #3,0:FOR i=0 TO 3:CURSOR #3,176-i,5-i:PRINT #3,Title$ 

144 INK #3,7:FOR i=0 TO 1:CURSOR #3,172+i,3  :PRINT #3,Title$ 

146 CSIZE #3,2,0:OVER #3,0 

148 END DEFine 

 

152 DEFine PROCedure Clk_Face 

154 INK #5,7:FILL #5,1:CIRCLE #5,48,50,35:FILL #5,0 

156 INK #5,3:FILL #5,1:CIRCLE #5,48,50,34:FILL #5,0 

158 INK #5,0 
160 FOR j=3 TO 5 STEP 2 

162    FOR i=0 TO 360 STEP j*12-30 

164       x=34*SIN(RAD(i)):y=34*COS(RAD(i)) 
166       x1=(34-j)*SIN(RAD(i)):y1=(34-j)*COS(RAD(i)) 

168       xx=x+48:xx1=x1+48:yy=y+50:yy1=y1+50 

170       LINE #5,xx,yy TO xx1,yy1 
172    END FOR i 

174 END FOR j 

176 INK #5,7 
178 CURSOR #5,100, 14:PRINT #5,'0' 

180 CURSOR #5,190, 78:PRINT #5,'15' 

182 CURSOR #5, 95,145:PRINT #5,'30' 
184 CURSOR #5,  1, 78:PRINT #5,'45' 

186 Time%=180:Sech%=180:Sec%=180:Countdown 

188 END DEFine 

192 DEFine PROCedure Menu 

194 STRIP #3,0:CSIZE #4,0,0:S=1:F=0:Clue%=0:Word%=1 

196 Command$=' Clue  Load  Play  Time  Word ' 

198 REPeat Comm_lp 

200  x=S*72+58:y=202:INK #3,7:CURSOR #3,58,y:PRINT #3,Command$:INK #3,0 

202  STRIP #3,4:CURSOR #3,x,y:PRINT #3,Command$(S*6+1 TO S*6+6) 

204  STRIP #3,0:CLS #4:Set_Word:Set_Clue 

206  STRIP #3,0:INK #3,4:k$=INKEY$(#4,30):k=CODE(k$) 

208  SELect ON k 

210   =192:S=S-1:IF S<0:S=4 
212   =200:S=S+1:IF S>4:S=0 

214   = 27:MODE 4:STOP 

216   = 10:SELect ON S 
218          =0:IF Clue%=1 AND Word%=1:Clue%=0:ELSE Clue%=1 

220          =1:Sel_Dev:CLS #4:S=2 

222          =2:IF F:count%=count%+1:Play:Score:ELSE S=1 

224          =3:Set_Time:CLS #4:S=2 

226          =4:IF Word%=1 AND Clue%=1:Word%=0:ELSE Word%=1 

228        END SELect 
230  END SELect 

232 END REPeat Comm_lp 

234 END DEFine 
 

238 DEFine PROCedure Set_Clue 

240 INK #3,2:IF Clue%=1:CURSOR #3,24,200:PRINT #3,'ON' 
242 INK #3,0:IF Clue%=0:CURSOR #3,24,200:PRINT #3,'  ' 

244 END DEFine 

 
248 DEFine PROCedure Set_Word 

250 INK #3,2:IF Word%=1:CURSOR #3,424,200:PRINT #3,'ON' 

252 INK #3,7:IF Word%=0:CURSOR #3,424,200:PRINT #3,'  ' 
254 END DEFine 

258 DEFine PROCedure Sel_Dev 

260 CLS #4:CURSOR #4,10,5:PRINT #4,'Select Device name <¿¾>:' 

262 REPeat Dev_lp 
264  CURSOR #4,300,5:PRINT #4,DR$(D%) 

266  k$=INKEY$(#4,20):k=CODE(k$) 

268  SELect ON k 
270   =208:D%=D%-1:IF D%<1:D%=6 

272   =216:D%=D%+1:IF D%>6:D%=1 

274   =10 :EXIT Dev_lp 
276   =27 :RETurn 

278  END SELect 

280 END REPeat Dev_lp 

282 SDR$=DR$(D%):FList 

284 END DEFine 

 
288 DEFine PROCedure FList 

290 n%=1:num%=20:ftot%=0 

292 DIM Volumn$(11),Sector$(24),file$(20,36),df$(36) 
294 CLS #4:CURSOR #4,10,5:PRINT #4,'Files being selected...' 

296 DELETE SDR$&'FList' 

298 OPEN_NEW #6,SDR$&'FList':DIR #6,SDR$:CLOSE #6 
300 OPEN_IN #6,SDR$&'FList':INPUT #6,Volumn$,Sector$ 

302 REPeat DIR_lp 

304  IF EOF(#6) OR n%>num%:ftot%=n%-1:CLOSE #6:EXIT DIR_lp 

306  INPUT #6,df$ 

308  IF 'Wrd_' INSTR df$<>0:file$(n%)=df$:n%=n%+1 

310 END REPeat DIR_lp 
312 IF ftot%<1 

314    CLS #4:CURSOR #4,10,5:PRINT #4,'No Word Files found...' 

316    PAUSE:CLS #4:RETurn 
318 END IF 

320 Sel_File 

322 END DEFine 

 

326 DEFine PROCedure Sel_File 

328 CLS #4:CURSOR #4,10,5:PRINT #4,'Select Wrd File <¾¿>:':n%=1 
330 REPeat File_lp 

332  str$=file$(n%,4+('Wrd_' INSTR file$(n%)) TO) 

334  CURSOR #4,270,5:PRINT #4,str$:CLS #4,4 
336  k$=INKEY$(#4,20):k=CODE(k$) 

338  SELect ON k 

340   =208:n%=n%-1:IF n%<1:n%=ftot% 
342   =216:n%=n%+1:IF n%>ftot%:n%=1 

344   =10 :EXIT File_lp 

346   =27 :RETurn 
348  END SELect 

350 END REPeat File_lp 

352 Wrd_Load 

354 END DEFine 

358 DEFine PROCedure Wrd_Load 

360 DIM Word$(100,14),Clue$(100,32):F=1:w%=1:CLS #4 

362 CURSOR #4,10,5:PRINT #4,'Loading Word File...' 

364 OPEN_IN #6,SDR$&file$(n%) 

366 REPeat Wrd_lp 
368  INPUT #6,Word$(w%),Clue$(w%) 

370  IF EOF(#6):CLOSE #6:EXIT Wrd_lp 

372  CURSOR #4,300,5:PRINT #4,w%:CLS #4,4:w%=w%+1 
374 END REPeat Wrd_lp 

376 END DEFine 

 
380 DEFine PROCedure Set_Time 

382 Sec%=Time%:Countdown 

384 CLS #4:CURSOR #4,10,5:PRINT #4,'Set Count Down Time: ¼ ½' 

386 REPeat Time_lp 

388  k$=INKEY$(#4,20):k=CODE(k$) 

390  SELect ON k 
392   =192:Sec%=Sec%+6:Countdown 

394   =200:Sec%=Sec%-6:Countdown 

396   = 10:Time%=Sec%:EXIT Time_lp 
398   = 27:RETurn 

400  END SELect 

402 END REPeat Time_lp 
404 END DEFine 

 

408 DEFine PROCedure Countdown 
410 INK #5,3 

412 LINE #5,48,50 TO 48+26*SIN(RAD(Sech%)),50+26*COS(RAD(Sech%)) 

414 INK #5,7:Sech%=Sec%:BEEP 200,10 
416 LINE #5,48,50 TO 48+26*SIN(RAD(Sech%)),50+26*COS(RAD(Sech%)) 

418 END DEFine 

422 DEFine PROCedure Play 

424 RANDOMISE:n%=RND(1 TO w%) 

426 IF Word$(n%)=key$ 
428    IF INKEY$(#4,5)=CHR$(27):RETurn :ELSE GO TO 424 

430 ELSE 

432    Wrd_Sort 
434 END IF 

436 Count=DATE:Sec%=Time%:Countdown 

438 ch%=1:pos%=1:key$=FILL$('_',lgth%) 
440 CURSOR #5,190,125:PRINT #5,key$:CLS #5,4 

442 REPeat Wrd_lp 

444  STRIP #5,5:CURSOR #5,174+16*ch%,125:PRINT #5,key$(ch%) 

446  k$=INKEY$(#4,20):k=CODE(k$) 

448  SELect ON k 

450   =192:ch%=ch%-1:IF ch%<1:ch%=1 
452   =200:ch%=ch%+1:IF ch%>lgth%:ch%=lgth% 

454   =32,48 TO 57,65 TO 90,97 TO 122 

456               IF Set(ch%)<>3:key$(ch%)=k$:ch%=ch%+1:ELSE ch%=ch%+1 
458               IF ch%>lgth%:ch%=1 

460  END SELect 

462  STRIP #5,1:CURSOR #5,174+16*pos%,125:PRINT #5,key$(pos%):pos%=ch% 
464  IF key$(ch%)==Word$(n%,ch%):Set(ch%)=3 

466  IF key$==Word$(n%) OR Sec%=0:EXIT Wrd_lp 

468  IF Count<>DATE:Sec%=Sec%-(6*(DATE-Count)):Countdown:Count=DATE 

470 END REPeat Wrd_lp 

472 FOR ch=1 TO LEN(key$):IF key$(ch)==Word$(n%,ch):per_score%=per_score%+5 

474 key$=Word$(n%):CURSOR #5,190,25:PRINT #5,key$ 
476 CSIZE #5,2,0:CURSOR #5,190,100:PRINT #5,'Score points: ';per_score%:CLS #5,4:CSIZE 

#5,3,1 

478 END DEFine 

482 DEFine PROCedure Wrd_Sort 

484 CSIZE #5,3,1:INK #5,7 

486 CURSOR #5,190,25:PRINT #5,FILL$(' ',15) 
488 Sort$=Word$(n%):lgth%=LEN(Sort$) 

490 DIM Set(lgth%) 

492 FOR Sort=1 TO lgth%-1 
494   r1=RND(1 TO lgth%):r2=RND(1 TO lgth%) 

496   Con1$=Sort$(r1):Con2$=Sort$(r2) 

498   Sort$(r1)=Con2$:Sort$(r2)=Con1$ 
500 END FOR Sort 

502 IF Sort$=Word$(n%):GO TO 492 

504 IF Word%=1:CURSOR #5,190,25:PRINT #5,Sort$ 

506 IF Word%=0 

508    FOR blk=1 TO lgth% 

510        BLOCK #5,14,20,174+blk*16,25,7 
512        BLOCK #5,10,18,176+blk*16,25+1,(blk MOD 8) 

514    END FOR blk 

516 END IF 
518 IF Clue%=1:CURSOR #4,10,5:PRINT #4,Clue$(n%):CLS #4,4 

520 max_score%=max_score%+5*lgth% 

522 CSIZE #5,2,0:CURSOR #5,190,55:PRINT #5,'Max   points: ';max_score%:CLS #5,4:CSIZE 
#5,3,1 

524 END DEFine 

 

528 DEFine PROCedure Score 

530 INK #5,3:FILL #5,1:CIRCLE #5,48,50,28:FILL #5,0 

532 INK #5,0:ARC #5,28,60 TO 44,60,-PI/2 
534          ARC #5,28,60 TO 44,60,-PI/2.3 

536          ARC #5,52,60 TO 68,60,-PI/2 

538          ARC #5,52,60 TO 68,60,-PI/2.3 
540 IF Sec%<6:ARC #5,30,40 TO 66,40,-PI/2:ARC #5,30,40 TO 66,40,-PI/2.2 

542 IF Sec%>0:ARC #5,30,40 TO 66,40,PI/2:ARC #5,30,40 TO 66,40,PI/2.2 

544 INK #5,1:FILL #5,1:CIRCLE #5,36,58,2.5:FILL #5,0 
546 INK #5,1:FILL #5,1:CIRCLE #5,60,58,2.5:FILL #5,0 

548 CLS #4:CURSOR #4,25,5:PRINT #4,'Press ENTER to continue...' 

550 k$=INKEY$(#4,-1):IF k$<>CHR$(10):GO TO 550 
552 INK #5,3:FILL #5,1:CIRCLE #5,48,50,28:FILL #5,0 

554 CURSOR #5,190,25:CLS #5,4 

556 CURSOR #5,190,125:CLS #5,4 
558 IF count%>=100 

560    CLS #4:CURSOR #4,25,5:PRINT #4,'Counters are to be reset...' 

562    PAUSE 100:count%=0:max_score%=0:per_score%=0 
564    CURSOR #5,190,50::CLS #5,4:CURSOR #5,190,95:CLS #5,4 

566 END IF 

568 END DEFine 

100 REMark QBWord_Gen (QBITS Word Generator •  1992) 

102 : 

104 Init:Title:Menu 

 

108 DEFine PROCedure Init 

110 MODE 4:OPEN #3,scr_512x256a0x0:PAPER #3,0:CLS #3:CLOSE #3 
112 OPEN #3,scr_480x230a16x4  :PAPER #3,32:CLS #3:CSIZE #3,2,0 

114 OPEN #4,con_444x34a34x220 :PAPER #4,0 :CLS #4:BORDER #4,1,2 

116 OPEN #5,con_128x164a38x38 :PAPER #5,0 :CLS #5:BORDER #5,2,0 
118 OPEN #6,con_264x164a208x38:PAPER #6,0 :CLS #6:BORDER #6,2,0 

120 CSIZE #4,1,0:INK #4,7:CSIZE #5,1,0:INK #5,7:CSIZE #6,1,0:INK #6,7 

122 DIM DR$(6,5),SDR$(5):RESTORE :FOR i=1 TO 6:READ DR$(i) 

124 DATA 'mdv1_','mdv2_','flp1_','flp2_','ram1_','ram2_' 

126 D=3:SDR$=DR$(3) 

128 END DEFine 
 

132 DEFine PROCedure Title 

134 DIM Title$(12):Title$=' QBWord_Gen ' 
136 BLOCK #3,480,15,0,0,0  :BLOCK #3,172,22,152,6,0 

138 BLOCK #3,168,22,152,2,4:CSIZE #3,3,1:OVER #3,1 

140 INK #3,0:FOR i=0 TO 3:CURSOR #3,144-i,2:PRINT #3,Title$ 
142 INK #3,0:FOR i=0 TO 3:CURSOR #3,144-i,5-i:PRINT #3,Title$ 

144 INK #3,7:FOR i=0 TO 1:CURSOR #3,140+i,3:PRINT #3,Title$ 

146 CSIZE #3,2,0 

148 INK #3,7:FOR i=0 TO 1:CURSOR #3,22+i,22 :PRINT #3,'Words' 

150 INK #3,7:FOR i=0 TO 1:CURSOR #3,392+i,22:PRINT #3,'Clues' 

152 OVER #3,0 
154 END DEFine 

 

158 DEFine PROCedure Menu 
160 STRIP #3,0:DIM Command$(36),k$(1):S=0:W%=0 

162 Command$=' New   Edit  Find  Load  Save  Quit ' 

164 REPeat Comm_lp 
166  x=S*72+24:y=202:INK #3,7 

168  CURSOR #3,24,y:PRINT #3,Command$:INK #3,0 

170  STRIP #3,7:CURSOR #3,x,y:PRINT #3,Command$(S*6+1 TO S*6+6) 
172  STRIP #3,0:k$=INKEY$(#4,30):k=CODE(k$) 

174  SELect ON k 

176   =192:S=S-1:IF S<0:S=5 
178   =200:S=S+1:IF S>5:S=0 

180   = 27:STOP 

182   = 10:SELect ON S 
184          =0:WNew                :CLS #4:S=1 

186          =1:IF W% :WEdit  :CLS #4:S=2 

188          =2:IF W% :WFind :CLS #4:S=1 
190          =3:DSel    :FList    :CLS #4:S=2 

192          =4:IF W% :DSel :WSave:CLS #4:S=0 

194          =5:STOP 
196        END SELect 

198  END SELect 

200 END REPeat Comm_lp 
202 END DEFine 

206 DEFine PROCedure WNew 

208 CURSOR #4,40,5:PRINT #4,'Create NEW WORD File (y/n) '; 

210 IF INKEY$(#4,-1)=='y' 
212    DIM Word$(96,15),Clue$(96,32),str$(36) 

214    CLS #5:CLS #6:W%=1:l=1:k=49:Pg_Prn 

216 END IF 
218 END DEFine 

 

222 DEFine PROCedure WEdit 

224 ch%=5:c%=1:Str_Chk:BCol%=7:ICol%=0:CURSOR #4,15,5 

226 PRINT #4,'(TAB) WORD/CLUE : (CTRL¼/½) DEL : (¾¿) UP/DOWN Line' 

228 REPeat Edit_lp 

230  CCol%=2:Str_Clr:Ln_Prn:Ln_Cur 

232  k$=INKEY$(#4,-1):k=CODE(k$) 

234  SELect ON k 
236   =  9:Ln_Prn:Str_Clr:c%=1 

238        IF ch%=5:ch%=6:ELSE ch%=5:END IF :Str_Chk 

240   = 10:BCol%=0:ICol%=7:Ln_Prn:Str_Clr:Wrd_Chk:RETurn 
242   =32 TO 126:Ln_Prn:Add_chr 

244   =194:CCol%=7:Ln_Cur:IF c%>1:c%=c%-1:Del_chr 

246   =202:CCol%=7:Ln_Cur:Del_chr 
248   =192:CCol%=7:Ln_Cur:IF c%>1:c%=c%-1 

250   =200:CCol%=7:Ln_Cur:IF c%<clen%+1:c%=c%+1 

252   =208:Ln_Clr:IF r%>0  :r%=r%-10:l=l-1:c%=1:Str_Chk 

254   =216:Ln_Clr:IF r%<150:r%=r%+10:l=l+1:c%=1:Str_Chk 

256  END SELect 

258 END REPeat Edit_lp 
260 END DEFine 

 

264 DEFine PROCedure WFind 

266 CURSOR #4,40,5 

268 PRINT #4,'Select WORD (¾¿) or PAGE (1 - 6)' 

270 REPeat Find_lp 
272  BCol%=7:ICol%=0:Ln_Prn 

274  k$=INKEY$(#5,-1):k=CODE(k$) 

276  SELect ON k 
278   =49 TO 54:Pg_Prn 

280   =208:BCol%=0:ICol%=7:Ln_Prn:IF r%>0  :r%=r%-10:l=l-1 

282   =216:BCol%=0:ICol%=7:Ln_Prn:IF r%<150:r%=r%+10:l=l+1 
284   = 10:BCol%=0:ICol%=7:Ln_Prn:RETurn 

286  END SELect 

288 END REPeat Find_lp 
290 END DEFine 

294 DEFine PROCedure Str_Chk 

296 IF ch%=5:clen%=LEN(Word$(l)):cmax%=15:str$=Word$(l) 

298 IF ch%=6:clen%=LEN(Clue$(l)):cmax%=32:str$=Clue$(l) 
300 END DEFine 

 

Checks on Word and Clue string lengths 
 

304 DEFine PROCedure Str_Clr 

306 IF LEN(str$)>cmax%:str$=str$(1 TO cmax%) 
308 IF ch%=5:Word$(l)=str$:ELSE Clue$(l)=str$ 

310 END DEFine 

 

Word and Clue boundaries for characters  

 

314 DEFine PROCedure Ln_Clr 
316 Str_Clr:BCol%=0:ICol%=7:CCol%=0:Ln_Prn:BCol%=7:ICol%=0:c%=1 

318 END DEFine 

 
Clear redundant Character form line 

 

322 DEFine PROCedure Ln_Prn 

324 STRIP #5,BCol%:INK #5,ICol% 

326 CURSOR #5,0,r%:PRINT #5,Word$(l)&FILL$(' ',15-LEN(Word$(l))) 

328 STRIP #6,BCol%:INK #6,ICol% 

330 CURSOR #6,0,r%:PRINT #6,Clue$(l)&FILL$(' ',32-LEN(Clue$(l))) 

332 END DEFine 

 
Line print 

 

336 DEFine PROCedure Pg_Prn 
338 INK #3,4:BCol%=0:ICol%=7:r%=-10 

340 FOR l=(k-48)*16-15 TO (k-48)*16:r%=r%+10:Ln_Prn 

342 CURSOR #3,22,5:PRINT #3,'Page ';k-48:r%=0:l=l-15 
344 END DEFine 

 

Page print 
 

348 DEFine PROCedure Ln_Cur 

350 BLOCK #ch%,8,1,c%*8-8,r%+9,CCol% 
352 END DEFine 

 

Underline cursor 

356 DEFine PROCedure Add_chr 

358 IF clen%<cmax% 

360    IF clen%>0 
362       IF c%=1:str$=k$&str$(1 TO) 

364       IF c%>1 AND c%<=clen% 

366          str$=str$(1 TO c%-1)&k$&str$(c% TO) 
368       END IF 

370       IF c%=clen%+1:str$=str$&k$ 

372       c%=c%+1:clen%=clen%+1 
374    END IF 

376    IF clen%=0:str$=k$:clen%=clen%+1:c%=c%+1 

378 ELSE 

380    IF c%<cmax%:str$(c%)=k$:c%=c%+1:ELSE str$(cmax%)=k$ 

382 END IF 

384 IF c%>cmax% OR clen%>cmax%:c%=cmax%:clen%=cmax% 
386 END DEFine 

 

390 DEFine PROCedure Del_chr 

392 IF clen%>1 

394    IF c%=1:str$=str$(2 TO clen%):clen%=clen%-1:END IF 

396    IF c%>1 AND c%<clen% 
398       str$=str$(1 TO c%-1)&str$(c%+1 TO):clen%=clen%-1:RETurn 

400    END IF 

402    IF c%>1 AND c%=clen% 

404       clen%=clen%-1:str$=str$(1 TO clen%) 

406    END IF 

408 ELSE 
410    str$="":clen%=0:c%=1 

412 END IF 

414 END DEFine 

416 : 

418 DEFine PROCedure Wrd_Chk 

420 IF Word$(l)<>'':str$=Word$(l):c=LEN(str$):ELSE RETurn 
422 FOR i=c TO 1 STEP -1:IF str$(i)=' ':str$=str$(1 TO c-1) 

424 END DEFine 

428 DEFine PROCedure DSel 

430 CLS #4:CURSOR #4,40,5 

432 PRINT #4,'Select Device    < ¿¾ >' 
434 REPeat Dev_lp 

436  CURSOR #4,230,5:PRINT #4,DR$(D) 

438  k$=INKEY$(#4,20):k=CODE(k$) 
440  SELect ON k 

442   =208:D=D-1:IF D<1:D=6 

444   =216:D=D+1:IF D>6:D=1 
446   = 10:SDR$=DR$(D):EXIT Dev_lp 

448  END SELect 

450 END REPeat Dev_lp 

452 END DEFine 

 

456 DEFine PROCedure FList 
458 DIM file$(20,36),Volumn$(12),Sector$(24),str$(36):f%=1:ftot%=0 

460 CLS #4:CURSOR #4,40,5:PRINT #4,'Files being selected...' 

462 DELETE SDR$&'FList' 
464 OPEN_NEW #8,SDR$&'FList':DIR #8,SDR$:CLOSE #8 

466 OPEN_IN #8,SDR$&'FList' 

468 INPUT #8,Volumn$,Sector$ 
470 REPeat DIR_lp 

472  IF EOF(#8) OR f%>20:ftot%=f%-1:CLOSE #8:EXIT DIR_lp 

474  INPUT #8,str$ 

476  IF 'Wrd_' INSTR str$<>0:file$(f%)=str$:f%=f%+1 

478 END REPeat DIR_lp 

480 IF ftot%<1 
482    CLS #4:CURSOR #4,40,5:PRINT #4,'No Word Files found...' 

484    PAUSE 100:CLS #4:S=0:RETurn 

486 END IF 
488 FSel 

490 END DEFine 

 
494 DEFine PROCedure FSel 

496 f%=1:CLS #4:CURSOR #4,40,5 

498 PRINT #4,'Select Word File < ¾¿ >' 
500 REPeat File_lp 

502  str$=file$(f%,4+('Wrd_' INSTR file$(f%)) TO) 

504  CURSOR #4,230,5:PRINT #4,str$:CLS #4,4 
506  k$=INKEY$(#4,20):k=CODE(k$) 

508  SELect ON k 

510   =208:f%=f%-1:IF f%<1:f%=ftot% 
512   =216:f%=f%+1:IF f%>ftot%:f%=1 

514   =10 :EXIT File_lp 

516   =27 :RETurn 
518  END SELect 

520 END REPeat File_lp 

522 WLoad 

524 END DEFine 

528 DEFine PROCedure WLoad 

530 DIM Word$(96,15),Clue$(96,32),str$(32):l=1:CLS #5:CLS #6 

532 CURSOR #3,22,5:PRINT #3,'      ' 
534 CLS #4:CURSOR #4,40,5:PRINT #4,'Loading Word File...' 

536 OPEN_IN #8,SDR$&file$(f%) 

538 REPeat Wrd_lp 
540  INPUT #8,Word$(l),Clue$(l) 

542  REMark CURSOR #4,220,5:PRINT #4,l:CLS #4,4 

544  IF EOF(#8)OR l=96:CLOSE #8:W%=1:EXIT Wrd_lp 
546  l=l+1 

548 END REPeat Wrd_lp 

550 l=1:k=49:Pg_Prn 

552 END DEFine 

 

556 DEFine PROCedure WSave 
558 CLS #4:INK #4,4:DIM str$(24):str$=SDR$&'Wrd_' 

560 CURSOR #4,100,20:PRINT #4,'<CTRL¼> Delete character  <ENTER> to quit' 

562 INK #4,7:CURSOR #4,20,5:PRINT #4,'Enter Filename ':Ed_Str 
564 IF 'Wrd_' INSTR str$=0 OR LEN(str$)<10:RETurn 

566 DELETE str$:OPEN_NEW #8,str$:chk=96 

568 FOR n=1 TO 96 
570  IF Word$(n)<>'':PRINT #8,Word$(n)\Clue$(n):chk=chk-1 

572 END FOR n 

574 CLOSE #8:IF chk=96:DELETE str$ 

576 END DEFine 

 

580 DEFine PROCedure Ed_Str 

582 lstr%=LEN(str$) 

584 REPeat Str_lp 

586 CURSOR #4,150,5:PRINT #4,str$:CLS #4,4 
588 lstr%=LEN(str$) 

590 k$=INKEY$(#4,-1):k=CODE(k$) 

592  SELect ON k 
594   =48 TO 57,65 TO 90,95,97 TO 122:IF lstr%<20:str$=str$&k$ 

596   =194:IF lstr%>5:str$=str$(1 TO 4)&str$(5 TO lstr%-1) 

598   = 10:RETurn 
600  END SELect 

602 END REPeat Str_lp 

604 END DEFine 

QBConundrum  
Having obtained a copy of QBConundrum SuperBASIC code and loaded it 

and the accompanying files into a recognised QL device. Use the QDOS 

command LRUN, as shown:-  
 

LRUN flp1_QBConundrum.  
  

All being well the Conundrum Game will direct you to select and load a 

Wrd_file, then test your knowledge of words against the clock face countdown. 

 
Notes on QL2K emulator  
Both the QLAY & QL2K emulators use an application tool to create a QDOS directory file and 

append or delete files in it. Creating a new qlay.dir file first open a Windows Command Prompt 
(Win 7 Press Start button in search programs and files box type command prompt: Win 10 in ask 

me anything box type command prompt.)  

Activate the command prompt window then navigate with DOS commands to the drive and 

Windows File Directory folder that holds your QL Files. 

 i.e C:\>chdir H:\QL\FDIR\WIN1_   H:\QLFDIR\WIN1_>dir 

This will list the files as a DOS directory. This needs to also contain a copy of QLAYT-86.EXE or 

QLAY-X64.EXE downloaded with QLAY or QL2K 

At the DOS prompt now enter this command: - 

i.e. H:\QL\FDIR\WIN1_>qlayt-x64.exe –c qlay.dir 

This should create a directory file qlay.dir to which you can now append files. For example:-  

i.e. H:\QL\FDIR\WIN1_>qlayt-x64.exe –i Boot 

This will append the File named ‘Boot’ to the qlay.dir .  

Once you have appended your files you can use the following command to list them:- 

i.e. H:\QL\FDIR\WIN1_> qlayt-x64.exe –l 

A list of files should now be shown contained within the qlay.dir 

 

 

 

 

 

 

 

 

To remove a file at the DOS prompt type:-  

i.e. H:\QL\FDIR\WIN1_>qlayt-x64.exe -r Golf 

This will remove the File named ‘Golf’ from the qlay.dir .  

 

Running the QL2K emulator the files listed in WIN1_ should now be readable by the QDOS DIR 
command; however these files will not be loadable or run if not compatible with the QDOS 

operating system you have. This even applies to QL software that does not work with or only works 

with certain versions of QL ROM’s or with added Toolkit extensions. 



 

 

 


